

Math 4530
Friday March 28
first and second fundamental forms in Maple

Here is a list of procedures to calculate first and second fundamental forms, as well as Gauss and Mean curvature, for regular parameterized surfaces. The procedures are modified from ones given on pages 119-121 of Oprea, mainly to use matrix algebra as we have done in class. They are illustrated with computations for the Enneper Surface, which was a hw problem from last week which was somewhat messy to do by hand.

```
> restart:
    #always a good idea
> with(linalg):
    #as much as possible we will use matrix algebra
Warning, the protected names norm and trace have been redefined and unprotected

> #dot product
dp:=proc(X,Y)
  X[1]*Y[1]+X[2]*Y[2]+X[3]*Y[3];
end:
>
> dp([1,2,3],[-1,2,1]);

6

> #2-norm
nrm:=proc(X)
  sqrt(dp(X,X));
end:
>
> nrm([1,2,3]);

 $\sqrt{14}$ 

> #cross product:
xp:=proc(X,Y)
  local a,b,c;
  a:=X[2]*Y[3]-X[3]*Y[2];
  b:=X[3]*Y[1]-X[1]*Y[3];
  c:=X[1]*Y[2]-X[2]*Y[1];
  [a,b,c];
end:
> xp([1,0,0],[0,2,0]);

[0,0,2]

> #Derivative matrix for mapping X:
DXq:=proc(X)
  local Xu,Xv;
  Xu:=matrix(3,1,[diff(X[1],u),diff(X[2],u),diff(X[3],u)]);
  Xv:=matrix(3,1,[diff(X[1],v),diff(X[2],v),diff(X[3],v)]);
  simplify(augment(Xu,Xv));
end:
> #Enneper surface parameterization, to play with:
```

```
Enne:=(u,v)->[u-u^3/3+u*v^2,v-v^3/3+v*u^2,u^2-v^2];
```

$$Enne := (u, v) \rightarrow \left[u - \frac{1}{3}u^3 + uv^2, v - \frac{1}{3}v^3 + vu^2, u^2 - v^2 \right]$$

```
> DXq(Enne(u,v));
```

$$\begin{bmatrix} 1-u^2+v^2 & 2vu \\ 2vu & 1-v^2+u^2 \\ 2u & -2v \end{bmatrix}$$

```
> #Matrix of first fundamental form:
```

```
gij:=proc(X)
```

```
local g11,g12,g22,Y;
```

```
Y:=evalm(DXq(X));
```

```
simplify(evalm(transpose(Y)*Y));
```

```
end:
```

```
> gij(Enne(u,v));
```

$$\begin{bmatrix} 2v^2u^2+1+2v^2+2u^2+v^4+u^4 & 0 \\ 0 & 2v^2u^2+1+2v^2+2u^2+v^4+u^4 \end{bmatrix}$$

```
> #unit normal:
```

```
N:=proc(X)
```

```
local Y,Z,s;
```

```
Y:=DXq(X);
```

```
Z:=xp(col(Y,1),col(Y,2));
```

```
s:=nrm(Z);
```

```
simplify(evalm((1/s)*Z));
```

```
end:
```

```
> N(Enne(u,v));
```

```
#if you look up csgn you see that it stands for
```

```
#"complex sign;" for real numbers it is +1 when
```

```
#the expression inside is positive and -1 when it
```

```
#is negative.
```

$$\left[-2 \frac{\text{csgn}((u^2+1+v^2)^2)u}{u^2+1+v^2}, 2 \frac{\text{csgn}((u^2+1+v^2)^2)v}{u^2+1+v^2}, -\frac{\text{csgn}((u^2+1+v^2)^2)(v^2+u^2-1)}{u^2+1+v^2} \right]$$

```
> #matrix of second fundamental form:
```

```
hij:=proc(X)
```

```
local Y,Xu,Xv,Xuu,Xuv,Xvv,U,h11,h12,h22;
```

```
Y:=DXq(X);
```

```
U:=N(X);
```

```
Xu:=col(Y,1);
```

```
Xv:=col(Y,2);
```

```
Xuu:=[diff(Xu[1],u),diff(Xu[2],u),diff(Xu[3],u)];
```

```
Xuv:=[diff(Xu[1],v),diff(Xu[2],v),diff(Xu[3],v)];
```

```
Xvv:=[diff(Xv[1],v),diff(Xv[2],v),diff(Xv[3],v)];
```

```

h11:=dp(Xuu,U);
h12:=dp(Xuv,U);
h22:=dp(Xvv,U);
simplify(matrix(2,2,[h11,h12,h12,h22]));
end:

```

```

>
>

```

```

> hij(Enne(u,v));

```

$$\begin{bmatrix} 2 \operatorname{csgn}((u^2 + 1 + v^2)^2) & 0 \\ 0 & -2 \operatorname{csgn}((u^2 + 1 + v^2)^2) \end{bmatrix}$$

```

>

```

```

> #matrix A of the (opposite) of the differential
#of the normal map
aij:=proc(X)
local Y,H,G;
H:=hij(X);
G:=gij(X);
simplify(evalm(inverse(G)*H));
end:

```

```

> aij(Enne(u,v));

```

$$\begin{bmatrix} 2 \frac{\operatorname{csgn}((u^2 + 1 + v^2)^2)}{2 v^2 u^2 + 1 + 2 v^2 + 2 u^2 + v^4 + u^4} & 0 \\ 0 & -2 \frac{\operatorname{csgn}((u^2 + 1 + v^2)^2)}{2 v^2 u^2 + 1 + 2 v^2 + 2 u^2 + v^4 + u^4} \end{bmatrix}$$

```

> #Gauss curvature
GK:=proc(X)
local A;
A:=aij(X);
det(A);
end:

```

```

> #Mean curvature
MK:=proc(X)
local A;
A:=aij(X);
trace(A);
end:

```

```

> GK(Enne(u,v));

```

$$-4 \frac{\operatorname{csgn}((u^2 + 1 + v^2)^2)^2}{(2 v^2 u^2 + 1 + 2 v^2 + 2 u^2 + v^4 + u^4)^2}$$

```

> MK(Enne(u,v));

```

[]

0